

```
1 ' -----
2 '     From the book Inside Active Directory, ISBN 0-201-61621-1
3 '     Copyright (C) 2002 by Addison-Wesley
4 '     Script by Sakari Kouti (see http://www.kouti.com)
5 ' You have a royalty-free right to use, modify, reproduce and distribute
6 ' this script (and/or any modified version) in any way you find useful,
7 ' provided that you agree that Addison-Wesley or Sakari Kouti has no
8 ' warranty, obligations or liability for the script. If you modify
9 ' the script, you must retain this copyright notice.
10 ' -----
11 Option Explicit
12
13 Const E_ADS_PROPERTY_NOT_FOUND = &H8000500D
14
15 Dim objDSE, objADObject, objClass, i, strPropName
16
17 Set objDSE = GetObject("LDAP://rootDSE")
18 Set objADObject = GetObject("LDAP://CN=Guest,CN=Users," & _
19     objDSE.Get("defaultNamingContext"))
20 objADObject.GetInfo
21
22 Set objClass = GetObject(objADObject.Schema)
23
24 i = 0      'this will count each property
25
26 For Each strPropName In objClass.MandatoryProperties
27     Call ShowPropAndValues(i, strPropName, objADObject)
28 Next
29
30 For Each strPropName In objClass.OptionalProperties
31     Call ShowPropAndValues(i, strPropName, objADObject)
32 Next
33
34 Sub ShowPropAndValues(i, strPropName, objADObject)
35     Dim objAttribute, strName
36
37     i = i + 1
38
39     Set objAttribute = GetObject("LDAP://schema/" & strPropName)
40     strName = i & ": " & strPropName & " (" & objAttribute.Syntax & ")"
41
42     Select Case objAttribute.Syntax
43         Case "OctetString", "ObjectSecurityDescriptor"
44             Call ShowBinary(strName, strPropName, objADObject)
45         Case "INTEGER8"
46             Call ShowLargeInteger(strName, strPropName, objADObject, _
47                 objAttribute)
48         Case Else
49             Call ShowOtherTypes(strName, strPropName, objADObject)
50     End Select
51 End Sub
52
53 Sub ShowBinary(strName, strPropName, objADObject)
54     Dim varPropValue, arrPropValues
55
56     On Error Resume Next
57
58     Err = 0
59     arrPropValues = objADObject.GetEx(strPropName)
60     If Err = E_ADS_PROPERTY_NOT_FOUND Then
61         varPropValue = "<not in property cache>"
62     Else
63         varPropValue = "???"
64     End If
65     Call ShowValue(strName, varPropValue)
66 End Sub
67
68 Sub ShowLargeInteger(strName, strPropName, objADObject, _
```

```
69     objAttribute)
70     Dim varPropValue, objLargeInt
71
72     On Error Resume Next
73     If objAttribute.MultiValued Then
74         varPropValue = "<cannot retrieve multivalued large integers>"
75     Else
76         Err = 0
77         Set objLargeInt = objADObject.Get(strPropName)
78         If Err = E_ADS_PROPERTY_NOT_FOUND Then
79             varPropValue = "<not in property cache>"
80         Else
81             varPropValue = objLargeInt.HighPart * 2^32 + _
82                 objLargeInt.LowPart
83         End If
84     End If
85     Call ShowValue(strName, varPropValue)
86 End Sub
87
88 Sub ShowOtherTypes(strName, strPropName, objADObject)
89     Dim arrPropName, varPropValue, arrPropValues
90
91     On Error Resume Next
92
93     Err = 0
94     arrPropValues = objADObject.GetEx(strPropName)
95     If Err = E_ADS_PROPERTY_NOT_FOUND Then
96         arrPropName = Array(strPropName)
97         Call objADObject.GetInfoEx(arrPropName, 0)
98         Err = 0
99         arrPropValues = objADObject.GetEx(strPropName)
100    End If
101    If Err = E_ADS_PROPERTY_NOT_FOUND Then
102        varPropValue = "<not set>"
103        Call ShowValue(strName, varPropValue)
104    Else
105        For Each varPropValue In arrPropValues
106            Call ShowValue(strName, varPropValue)
107        Next
108    End If
109 End Sub
110
111 Sub ShowValue(strName, varPropValue)
112     Dim intGap      'number of spaces before the value
113
114     intGap = 40 - Len(strName)
115     If intGap < 1 Then intGap = 1
116     WScript.Echo strName & String(intGap, " ") & varPropValue
117     strName = ""
118 End Sub
119
```